

# Variációk egy témára - táblázatkezelő feladatok megoldása többféleképpen

Fehérné Mázsár Gabriella  
(femaga@index.hu)

A gondolkodás fejlesztésének a programozás mellett a másik nagyon hatékony lehetősége a táblázatkezelő feladatok megoldása. A számítástechnika órákon a felmerülő feladatokat a tanulók többféleképpen is megoldhatják: különböző függvények alkalmazásával, esetleg irányított szűrés segítségével.

Cikkünkben egy statisztikai jellegű (iskolai életből vett) feladatsor változatos megoldási módzatait szeretnénk bemutatni Excel segítségével.

A feladatok kitűzésekor vegyük figyelembe, hogy a diákok a függvények alkalmazását milyen mélységig sajátították el és milyen matematikai ismeretekkel rendelkeznek.

A megoldandó feladat:

*Készíts egy táblázatot, mely egy osztály tanulóinak érdemjegyeit tartalmazza irodalomból, nyelvtanból, matekból, kémiából, történelemből és németből; valamint az igazolt és az igazolatlan mulasztások számát, tanulónként.*

a) *Készíts statisztikát, mely tartalmazza a következőket:*

- 1. a kitűnő tanulók száma*
- 2. a bukott tanulók száma*
- 3. az összes bukás száma*
- 4. azoknak a tanulóknak a számát, akik 10 óránál többet mulasztottak igazolatlanul*
- 5. azoknak a tanulóknak a számát, akik egyetlen órát sem hiányoztak*

b) *Egészítsd ki a táblázatot egy Megjegyzés és egy Figyelmeztetés nevű oszloppal!*

- 1. Az előbbiben, a legtöbbet mulasztó tanuló sorában szerepeljen a szöveg "Legtöbbet mulasztó!"*
- 2. Figyelmeztetés azoknak a sorába kerül, akik 10 óránál többet hiányoztak igazolatlanul, ennek szövege: "A szülőt értesíteni!"*

## Megoldások:

a)

### 1. A kitűnő tanulók száma:

#### 1. megoldás:

*Kitűnő tanuló, akinek minden tantárgyból ötöse van.*

A kitűnő tanulók számát megkapjuk, ha egy új oszlopban kiszámoljuk a tanulók tantárgyi átlagát, majd egy cellába beírjuk a DARABTELI függvényt, mely a tantárgyi átlag oszlopban megszámlálja az ötösök számát. (A tantárgyi átlagot egy tizedes pontossággal számolja!)

	A	B	C	D	E	F	G	H	I	J
1	Név	Irodalom	Nyelvtan	Maték	Kémia	Történelem	Német	Tantárgyi átlag		
2	Bandl Katalin	3	2	4	3	5	5	3,7	=ÁTLAG(B2:G2)	
3	Fehér Ágnes	5	4	2	5	1	3	3,3		
4	Fekete Péter	4	3	1	4	2	2	2,7		
5	Görög Géza	3	4	2	1	1	3	2,3		
6	Hollenbach Éva	3	2	3	4	4	5	3,5		
7	Kis Béla	1	1	2	2	3	2	1,8		
8	Magyar Balázs	5	1	4	4	2	5	3,5		
9	Mázsás Dezső	3	1	4	2	1	4	2,5		
10	Nagy Ervin	5	5	5	5	5	5	5,0		
11	Nyisztor Attila	4	5	5	1	5	3	3,8		

Ötösök száma: =DARABTELI(H2:H11;"5")

#### 2. megoldás:

1. lépés: A tantárgyi átlag oszlopa helyett vegyük fel a Kitűnő oszlopot. A H2 cellába írjuk be a következő függvényt: =HA(ÉS(B2=5;C2=5;D2=5;E2=5;F2=5;G2=5);"kitűnő";""). Az ÉS függvényt használva, ha az összes tantárgyból ötös osztályzatot kap a tanuló, akkor a HA függvény igaz ága teljesül és a cellában a kitűnő felirat jelenik meg. Ha valamelyik tantárgyból az osztályzata nem lett ötös, akkor nem jeleneik meg semmilyen felirat a cellában.

Az H2-es cellában lévő függvényt húzzuk le a kitöltő fogantyúval a H11-ig!

2. lépés: Egy tetszőleges cellába írjuk be a következő függvényt: =DARABTELI(H2:H11;"kitűnő")

A beírt DARABTELI függvénnyel számoljuk meg, hogy a H2:H11 tartomány hány darab "kitűnő" feliratot tartalmaz!

#### 3. megoldás:

Irányított szűrővel szűrjük ki azoknak a nevének, akik kitűnő tanulók.

Ha több feltételt adunk meg a különböző oszlopokhoz, akkor az összes feltételt a feltéltartomány azonos sorába írjuk be. Például ez a feltéltartomány azokat a sorokat jeleníti meg, amelyekben az irodalom, a nyelvtan, a matek, a kémia, a történelem és a német oszlopban 5 szerepel.

Irodalom	Nyelvtan	Matek	Kémia	Történelem	Német
5	5	5	5	5	5

A fenti táblázat a szűrőtartomány táblázata.

A listatartomány: A1:G11 tartománya.

Szűrés előtt írjuk be a Név szót abba a cellába, ahová szűrni szeretnénk a listatartományból a szűrőtartománynak megfelelő neveket!

Eredmény:

Név
Nagy Ervin

DARAB2 függvénnyel adjuk össze hány darab szó szerepel a kiszűrt nevek tartományában!

A kiszűrt nevek tartományát nevezzük el "kitűnő" tartománynévvvel!

=DARAB2(kitűnő) függvény a kitűnő tanulók számát adja eredményül (1).

## 2. Bukott tanulók száma:

### 1. megoldás:

Vegyünk fel egy új oszlopot, melynek neve: Bukott! Ez az oszlop legyen az I!

Ha függvénnyel ennek az oszlopnak a celláiban kiíratjuk a "bukott" szót, a tanulók bukása esetén.

*Bukottnak számít az a tanuló, aki legalább egy tantárgyból egyes osztályzatot kap.*

A VAGY függvény igaz logikai értéket ad eredményül, ha valamelyik feltétel igaz.

Írjuk be az I2-es cellába a következő függvényt:

=HA(VAGY(B2=1;C2=1;D2=1;E2=1;F2=1;G2=1);"bukott";"")

A VAGY függvényt használva, ha legalább az egyik tantárgyból egyes osztályzatot kap a tanuló a feltétel igaz ága teljesül: a "bukott" szó jelenik meg a cellában.

Ha egyik tantárgyból sem kap egyes osztályzatot, a feltétel hamis ága érvényes: a cella nem tartalmaz felíratot.

Az I2-es cellában lévő függvényt húzzuk le a kitöltő fogantyúval az I11-ig!

Egy cellában DARABTELI függvénnyel számoljuk meg az I2:I11 tartomány "bukott" feliratainak a számát!

=DARABTELI(I2:I11;"bukott") Így megkapjuk a bukott tanulók számát (7).

## 2. megoldás:

Vegyünk fel egy új oszlopot! Ez legyen a J oszlop, melynek címe "Egyesek száma"!

A J2 cellába írjuk be a következő függvényt:

=DARABTELI(B2:G2;"1")

A J2 cella függvényét húzzuk le a kitöltő fogantyúval a J11-ig! Eredmény: megkapjuk az egyes tanulók egyeseinek számát.

Egy cellába írunk be ismét egy DARABTELI függvényt, mely a J2:J11 tartományban a 0-nál nagyobb számot tartalmazó cellák darabszámát adja meg.

=DARABTELI(J2:J11;">0") Eredmény: a bukott tanulók száma (7).

## 3. megoldás:

Írányított szűrővel szűrjük ki azoknak a nevét, akik megbuktak valamelyik tantárgyból!

Ha különböző oszlopokhoz különböző feltételek csoportját adjuk meg, a feltételeket külön sorokba kell beírunk. Például ez a feltételtartomány azokat a sorokat jeleníti meg, amelyekben az irodalom oszlopban 1, vagy a nyelvtan oszlopban 1 szerepel, vagy a matek oszlopban 1, vagy a kémia oszlopban 1, vagy a történelem oszlopban 1, vagy a német oszlopban 1 szerepel.

Irodalom	Nyelvtan	Matek	Kémia	Történelem	Német
1					
	1				
		1			
			1		
				1	
					1

A fenti táblázat a szűrőtartomány táblázata.

A listatartomány: A1:G11 tartománya.

Szűrés előtt írjuk be a Név szót abba a cellába, ahová szűrni szeretnénk a listatartományból a szűrőtartománynak megfelelő neveket!

Eredmény:

Név
Fehér Ágnes
Fekete Péter
Görög Géza
Kis Béla
Magyar Balázs
Mázsás Dezső
Nyisztor Attila

DARAB2 függvényel számoljuk meg hány szó szerepel a kiszűrt nevek tartományában!

A kiszűrt nevek tartományát nevezzük el "nevek" tartomány-névvvel!

=DARAB2(nevek) függvény a bukott tanulók számát adja

eredményül.

### 3. Összes bukás száma:

#### 1. megoldás:

DARABTELI függvénnyel számoljuk meg a B2:G11 tartományban hány 1-es van összesen!

=DARABTELI(B2:G11;"1")

Eredmény: megkapjuk, hogy az osztályban összesen hány bukás fordult elő.

#### 2. megoldás:

12							
13							
14	<b>egyesek száma:</b>	1	3	1	2	3	0
15							
16		<b>Irodalom</b>	<b>Nyelvtan</b>	<b>Matek</b>	<b>Kémia</b>	<b>Történelem</b>	<b>Német</b>
17		1	1	1	1	1	1
18							
19	<b>bukások száma:</b>	10					
20							

AB.DARAB függvénnyel adjuk meg tantárgyanként az egyesek számát. Ehhez adjuk meg tantárgyanként a feltételt. Például az irodalom tantárgy esetében a feltétel a B16:B17 tartományban található.

Álljunk a B14-es cellába. Az AB.DARAB függvény első argumentumaként adjuk meg abszolút hivatkozással a tanulók nevét, tantárgyait és azokhoz tartozó jegyeket tartalmazó adatok tartományát: \$A\$1:\$G\$11; második argumentumaként adjuk meg azt a mezőt cellahivatkozással, amelyikre ki szeretnénk számolni az egyesek számát. Ez jelen esetben az "Irodalom" mező, melyet cellahivatkozással megadva B1-el jelölünk. Az AB.DARAB függvény harmadik argumentuma legyen a feltétel tartomány: B16:B17.

A kitöltő fogantyúval húzzuk a B14 cellában lévő függvényt a G14-ig!

Álljunk a B19-es cellába! SZUM függvénnyel számoljuk ki a bukások számát. Írjuk be a következő függvényt: =SZUM (B14:G14) A beírt függvény összegzi az argumentumlistájában lévő számokat.

### 4. 10 óránál többet hiányoztak igazolatlanul:

A K oszlop legyen az „Igazolatlan mulasztások” oszlopa!

K
Igazolatlan mulasztások
13
0
7
16
0
15
0
0
6
10

### 1. megoldás:

Az igazolatlan mulasztások oszlopában (K oszlop) adjuk össze DARABTELI függvénnyel azokat a cellákat, amelyekben tíznél nagyobb szám szerepel!

=DARABTELI(K2:K11;">10")

Eredmény: 3.

### 2. megoldás:

Egy üres helyre vegyük fel a következő feltételtartományt (például legyen ez a tartomány a C21:C22).

Igazolatlan mulasztások
>10

AB.DARAB függvénnyel adjuk meg, hogy az Igazolatlan mulasztások oszlopban hányszor fordul elő 10-nél nagyobb szám.

Egy üres cellába írjuk be a következő függvényt: =AB.DARAB(A1:K11;11;C21:C22).

A függvény első argumentuma az adatbázisunk, a második argumentuma az Igazolatlan mulasztások oszlopa (a 11. oszlop), a harmadik argumentuma a feltételtartomány.

## 5. Egyetlen órát sem hiányzott tanulók száma:

### 1. megoldás:

Az L oszlop neve legyen „Igazolt mulasztások”.

Adjuk össze egy új oszlopban (M oszlop) az igazolatlan és az igazolt mulasztások számát! Az új oszlop neve legyen "Mulasztások összege"!

K	L	M	N	O
Igazolatlan mulasztások	Igazolt mulasztások	Mulasztások összege		
13	0	13	=SZUM(K2:L2)	
0	78	78		
7	34	41		
16	130	146		
0	45	45		
15	132	147		
0	0	0		
0	65	65		
6	149	155		
10	99	109		

Az M2 cellába írjuk be a következő SZUM függvényt a mulasztások összeadásához: =SZUM(K2:L2). A kitöltő fogantyúval húzzuk le az M2 cellában

lévő függvényt az M11-ig!

DARABTELI függvénnyel számoljuk meg hány darab 0 van az M2:M11 tartományban:

=DARABTELI(M2:M11;"=0") Eredmény: 1.

## 2. megoldás:

Az előbb leírt módon számítsuk ki a tanulók mulasztásainak összegét, majd vegyük fel a táblázatunk alatt a következő feltételtartományt (például legyen ez a tartomány a C24:C25).

<b>Mulasztások összege</b>
□

AB.DARAB függvénnyel adjuk meg, hogy a Mulasztások összege oszlopban hány darab 0 fordul elő. Egy üres cellába írjuk be a következő függvényt: =AB.DARAB(A1:M11;13;C24:C25). A beírt függvény első argumentuma az adatbázisunk, a második argumentuma a Mulasztások összege oszlopa (a 13. oszlop), a harmadik argumentuma a feltételtartomány.

b)

### 1. A legtöbbet mulasztó tanuló sorában szerepeljen a szöveg "Legtöbbet mulasztó!"

Egészítsük ki a táblázatot egy Megjegyzés nevű oszloppal (N oszlop)!

=HA(MAX(M\$2:M\$11)=M2;"Legtöbbet mulasztó";"") függvényt írjuk az N2 cellába és a kitöltő fogantyúval húzzuk le a függvényt az N2 cellától az N11 celláig!

MAX függvénnyel a "Mulasztások összege" oszlopból keressük meg a legnagyobb számot! A HA függvény igaz ága teljesülése esetén kiírja a "Legtöbbet mulasztó!" megjegyzést.

### 2. Figyelmeztetés azoknak a sorába kerül, akik 10 óránál többet hiányoztak igazolatlanul, ennek szövege: "A szülőt értesíteni!"

Egészítsük ki a táblázatot egy Figyelmeztetés oszloppal (O oszlop)!

Az O2 cellába írja be a következő függvényt:

=HA(K2>10;"A szülőt értesíteni!";"") és a kitöltő fogantyúval húzzuk le a függvényt az O2 cellától az O11 celláig!

N	O
<b>Megjegyzés</b>	<b>Figyelmeztetés</b>
=HA(MAX(M\$2:M\$11)=M2;"Legtöbbet mulasztó";"")	=HA(K2>10;"A szülőt értesíteni!";"")

## Ajánlott irodalom:

Baróti Ernőné: Hogyan használjuk?...EXCEL 5.0, 7.0, LSI Oktatóközpont, Budapest, 1998.

Greznér Ferenc: Excel 97 hatszor hat órában, Szak Kiadó Kft., Bicske, 1998.

Falus Iván-Ollé János: Statisztikai módszerek pedagógusok számára, Okker Kiadó, Budapest, 2000.

Krnács András-Kiss Csaba: Táblázatkezelési ismeretek, Műszaki Könyvkiadó, Budapest, 1999.

Szalayné Tahy Zsuzsa: Feladatgyűjtemény a táblázatkezeléshez, Nemzeti Tankönyvkiadó, Budapest, 1998.